

```

1  ! VEMA 2.gyak
2  ! Síkbeli rácsos tartó
3  ! [N,mm,MPa]
4
5  FINISH      ! kilépés minden folyamatból
6  /CLEAR      ! memória törlése
7
8  /CWD, 'D:\Google_Drive\01_education\05A_VEMA\gyakok21221\gyak2' ! munkakönyvtár beállítása
9  *ABBR,START,/input,truss2D,ans ! START gomb létrehozása ami beolvassa a truss2D.ans fájlt
10
11 /PNUM,kp,1   ! keypoint számozás bekapcsolás
12 /PNUM,line,1 ! line számozás bekapcsolás
13
14 ! ADATBEVITEL
15 a=5000      ! méret megadás
16 h=10000     ! magassági méret megadás
17
18 n1=3        ! jobbra dőlő blokkok száma
19 n2=3        ! balra dőlő blokkok száma
20
21 Eacel=200000 ! rugalmassági modulus
22 Akm=706.86  ! keresztmetszet terulete
23
24 ! PREPROCESSING !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
25 /PREP7      ! belépés a preprocesszáló részbe
26
27 ! GEOMETRIA MEGRAJZOLÁSA:
28 K,1,0,0     ! bal alsó keypoint megadása
29 K,2,0,h     ! bal felső keypoint megadása
30 L,1,2       ! előbbi két pont összekötése egy egyenessel
31
32 ! változók kezdeti értékeinek megadása
33 p1=1        ! keypoint sorszám
34 p2=2        ! keypoint sorszám
35 x1=0        ! kezdeti x koordináta
36
37 *DO,i,1,n1  ! do ciklus a 3 baloldali (jobbra dőlő) blokk létrehozására
38   p01=p1    ! bal alsó keypointok sorszáma -> 1:p01=1 2:p01=3 3:p01=5
39   p02=p2    ! bal felső keypointok sorszáma -> 1:p02=2 2:p02=4 3:p02=6
40   p1=p1+2   ! jobb alsó keypoint sorszám növelése -> 1:p1=3 2:p1=5 3:p1=7
41   p2=p2+2   ! jobb felső keypoint sorszám növelése -> 1:p2=4 2:p2=6 3:p2=8
42   x1=x1+a   ! jobb oldali keypointok x koordinátái -> 1:x1=5000 2:x1=10000 3: x1=15000
43   K,p1,x1,0 ! jobb alsó keypointok létrehozása -> 1:K,3,5000,0 2:K,5,10000,0 3:K,7,15000,0
44   K,p2,x1,h ! jobb felső keypointok létrehozása -> 1:K,4,5000,10000 2:K,6,10000,10000 3:K,8,15000,10000
45   L,p01,p1  ! alsó vízszintes vonalak megrajzolása -> 1:L,1,3 2:L,3,5 3:L,5,7
46   L,p02,p2  ! felső vízszintes vonalak megrajzolása -> 1:L,2,4 2:L,4,6 3:L,6,8
47   L,p1,p2   ! jobb oldali függőleges vonalak megrajzolása -> 1:L,3,4 2:L,5,6 3:L,7,8
48   L,p01,p2  ! átlós vonalak megrajzolása -> 1:L,1,4 2:L,3,6 3:L,5,8

```

```

49 *ENDDO      ! do ciklus lezárása
50
51 *DO,i,1,n2  ! do ciklus a 3 jobboldali (balra dőlő) blokk létrehozására
52   p01=p1    ! bal alsó keypointok sorszáma -> 1:p01=7 2:p01=9 3:p01=11
53   p02=p2    ! bal felső keypointok sorszáma -> 1:p01=8 2:p01=10 3:p01=12
54   p1=p1+2   ! jobb alsó keypoint szorzám növelése -> 1:p1=9 2:p1=11 3:p1=13
55   p2=p2+2   ! jobb felső keypoint szorzám növelése -> 1:p2=10 2:p2=12 3:p2=14
56   x1=x1+a   ! jobb oldali keypointok x koordinátái -> 1:x1=20000 2:x1=25000 3: x1=30000
57   K,p1,x1,0 ! jobb alsó keypointok létrehozása -> 1:K,9,20000,0 2:K,11,25000,0 3:K,13,30000,0
58   K,p2,x1,h ! jobb felső keypointok létrehozása -> 1:K,10,20000,10000 2:K,12,25000,10000 3:K,14,30000,10000
59   L,p01,p1  ! alsó vízszintes vonalak megrajzolása -> 1:L,7,9 2:L,9,11 3:L,11,13
60   L,p02,p2  ! felső vízszintes vonalak megrajzolása -> 1:L,8,10 2:L,10,12 3:L,12,14
61   L,p1,p2   ! jobb oldali függőleges vonalak megrajzolása -> 1:L,9,10 2:L,11,12 3:L,13,14
62   L,p02,p1  ! átlós vonalak megrajzolása -> 1:L,8,9 2:L,10,11 3:L,12,13
63 *ENDDO      ! do ciklus lezárása
64
65 ! ANYAGTULAJDONSÁGOK:
66 MP,ex,1,Eacel      ! rugalmassági modulus
67
68 ! HÁLÓZÁS:
69 ET,1,link180        ! elem típus: link180 (húzott-nyomott rúdelem)
70 SECTYPE,1,link      ! keresztmetszet típus: link (húzott-nyomott keresztmetszet)
71 SECDATA,Akm         ! keresztmetszet jellemzők (keresztmetszet területe)
72 LESIZE,all,,1       ! rúdelem méret megadása (rúd sorszáma:all->összes rúd,méret-,szögméret-,elemek száma vonalanként)
73 LMESH,all           ! összes vonal behálózása link180-as elemekkel
74
75 ! PEREMFELTÉTELEK:
76 D,node(0,0,0),all,0 ! 0,0,0 koordinátájú csomópont minden irányú megfogása
77 D,node((n1+n2)*a,0,0),uy,0 ! 30000,0,0 koordinátájú csomópont y irányú megfogása
78 F,node(0,h,0),fx,20000 ! 0,10000,0 koordinátájú csomópontra 20000N x irányú erő megadása
79 F,node(n1*a,0,0),fy,-20000 ! 15000,0,0 koordinátájú csomópontra -20000N y irányú erő megadása
80 F,node((n1+1)*a,0,0),fy,-40000 ! 20000,0,0 koordinátájú csomópontra -40000N y irányú erő megadása
81
82 ALLSEL            ! minden kiválasztása
83 FINISH            ! kilépés a preprocessingből
84
85 ! SOLVING !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
86 /SOLU             ! belépés a solver-be
87 SOLVE             ! számítás elvégzése
88 FINISH            ! kilépés a solver-ből
89
90 ! POSTPROCESSING !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
91 /POST1            ! belépés a postprocessor-ba
92
93 PLDISP            ! elmozdulás szemléltetése
94 PLNSOL,u,sum      ! elmozdulásmező abszolútértékének szemléltetése
95 PLNSOL,u,x        ! x irányú elmozdulásmező szemléltetése
96 PLNSOL,u,y        ! y irányú elmozdulásmező szemléltetése

```

```
97
98 ! feszültségek ábrázolása:
99 ETABLE,Sxxi,ls,1      ! szigma x rúdirányú normálfeszültségek a rudak i-edik csomópontjában -> Sxxi
100 ETABLE,Sxxj,ls,2      ! szigma x rúdirányú normálfeszültségek a rudak j-edik csomópontjában -> Sxxj
101 PLLS,Sxxi,Sxxj        ! szigma x rúdirányú normálfeszültségek ábrázolása
102 ! rúderők ábrázolása:
103 ETABLE,N,smisc,1      ! rúderő az egyes rudakban -> N
104 PLLS,N,N              ! rúderők ábrázolása
105
106 PRRSOL                ! támasztóerők listázása
```